NBSIR 75-699

The Calibration of Photographic Edges at NBS

Richard E. Swing

Institute for Basic Standards
National Bureau of Standards
Washington, D. C. 20234

April 22, 1975

Interim Report

U.S. DEPARTMENT OF COMMERCE
NATIONAL BUREAU OF STANDARDS

NBSIR 75-699

THE CALIBRATION OF PHOTOGRAPHIC EDGES AT NBS

Richard E. Swing

Institute for Basic Standards
National Bureau of Standards
Washington, D. C. 20234

April 22, 1975

Interim Report

U.S. DEPARTMENT OF COMMERCE, Rogers C.B. Morton, Secretary
NATIONAL BUREAU OF STANDARDS, Richard W. Roberts, Director

This document has been prepared as part of the program to upgrade and improve the density-measurement capability at NBS through research in the Optics and Micrometrology Section of the Optical Physics Division. It is the second of several that will be released in this subject-area, and will occasionally undergo revision and expansion as the need arises. At the end of the program, a complete NBS report on Densitometry and related matters will be prepared and published.

INTRODUCTION

NBS has produced photographic edges for optical and photographic emulsion testing for more than ten years, but has lacked a quantitative analytical procedure to certify their quality. The technique used to make these edges employs x-ray exposure of high resolution plates (with a thin tantalum strip for the edge-discontinuity), and their subsequent chemical processing is aimed at preventing adjacency effects. Because of the available x-ray equipment, exposure quality and uniformity have not always been consistent, and a need for quantitative evaluation of the edge has always existed.

Past evaluative procedures traced the edge on the NBS microdensitometer (a Kodak Model III, modified)* and inspected the obvious characteristics of the chart record such as steepness of slope, symmetry, exposure or processing variations manifested by the lack of smoothness in the trace and the variation in density on the toe and shoulder (to detect possible adjacency effects), always in a qualitative manner. The present note is concerned with computer analysis of the traces to quantify the statements made about the edges and their quality.

THE ANALYSIS OF EDGES

The major parameter used to characterize the edge is called acutance. This is a property that relates to subjective (visual) judgements of "sharpness," and is discussed in more detail in Ref. 1. It is a calculation of the mean-square slope of an edge, between two points along it, corrected (or adjusted) for the density difference between these points. This note documents a computer program called ACUMTF that calculates this quantity.

A further consideration of this note is the use of edges to test or evaluate an optical instrument through a procedure known as edge-gradient analysis. When the edge is placed in a microdensitometer and scanned, the resulting chart record can be analyzed to determine the optical transfer function of the microdensitometer. There are other variations to edge-analysis, but this is the only one considered in this note. To carry out these calculations, a second program is needed. It is called EMTFER and this note also documents that program.

---

*Certain commercial instruments and equipment are identified in this report in order to specify the experimental procedure adequately. In no case does such identification imply recommendation or endorsement by NBS, nor does it imply that the equipment identified is necessarily the best available for the purpose. Further, the equipment may not be identical to those models in current production that bear nominally similar designations of model type.

Finally, because a plot of the modulation transfer function (MTF) is often useful for diagnostic and display purposes, a program that plots the MTF produced by EMTFER has been prepared. It is called MTFPLT, and this note documents that program as well.

All three programs are written in BASIC, a language often available on the time-sharing computers of the kind used at NBS by the Optics and Micrometrology Section. Such systems are adequate for the task, are easily used by the inexperienced, the costs are low and the turn-around-time is short. For the accuracy commensurate with the analyses described in this note, use of a more sophisticated computer with a higher-level programming language would not be cost-effective and would probably provide a computational "overkill." Since these programs might also be useful to those outside NBS, it was felt that the BASIC language would offer the least problem in adaptation to other computers. These three program are used at NBS, exactly as written and listed in this note, to calibrate and examine photographic edges made there.

The various plots and figures of this note all derive from the same density calibration and edge-trace information located in the data section of ACUMTF.

ACUMTF

The program ACUMTF calculates acutance from a microdensitometer trace of an edge. Acutance is defined analytically by

$$\text{Acutance} = \frac{\frac{1}{n} \sum_{i=1}^{n} (\Delta D/\Delta x)_i^2}{D_b - D_a} . \qquad (1)$$

In this equation, $D_b - D_a$ is the density range over which the actual calculations are made. The end-points of the calculation are critical; if too much of the edge trace is included, the value of acutance will be too low, while the opposite it true if too little is included. It was found (Ref. 1) that the end points should be located on that portion of the trace where the slope $(\Delta D/\Delta x) = 0.005$, with x measured in micrometres. The accuracy of the calculations is improved when n, the number of $(\Delta D/\Delta x)$'s, is large.

Originally, acutance was devised to measure an attribute of photographic images that would correlate well with subjective impressions of image "sharpness." Thus, the image of an edge of the photographic material was analyzed and the property, acutance, ascribed to the material itself. The higher the value of acutance, the higher the corresponding quality. However, if the edge itself is traced, an analogous acutance can be calculated for a given set of microdensitometer parameters. This acutance applies only to the given set of parameters, and constitutes a calibration of sorts when those conditions can be duplicated on another instrument. When edges of extremely high acutance (values of 100,000 and larger) are used to test an optical instrument, no significant correction need be applied for the inherent imperfections of the edge, unless the response of the instrument itself approaches the quality of the edge. The program ACUMTF is intended for use in calibrating the edges made at NBS. When reported, values of acutance are always accompanied by details of the optics of the microdensitometer making the trace. Because of the problems of alignment of the scanning slit with respect to the edge, acutance values are always the average of at least three separate determinations. A listing of ACUMTF is shown in Appendix A.

Data Input to ACUMTF

The basic data for ACUMTF are:

1) A density calibration, in the form of diffuse density versus chart reading, and

2) An edge-trace, in the form of chart reading versus distance across the edge, taken from the chart recording produced by the (NBS) Kodak Model III microdensitometer.

Standard instrument calibration procedures are employed. A step tablet of the same material as that of the edge (and usually processed with it) whose density values are determined on a macrodensitometer measuring diffuse (visual) density is scanned in the microdensitometer. The values of chart reading versus diffuse (visual) density are then input on lines 900 through 915. Line 900 always contains the number of pairs of points, while the succeeding lines contain the point-pairs:

a) List chart reading and density, in that order, in pairs, in increasing order. It is important that both density and chart reading be listed in increasing order; the program will not provide a correct calibration table otherwise. See lines 910 through 914 of the ACUMTF listing for a specific example.

b) Do not interpolate visually on the chart recording to other than one-half a scale division; i.e., a value of 15.3, say, cannot be entered: it must be either 15.0 or 15.5. This is consistent with the inherent accuracy of the acutance calculations and the density calibration of the program, and will carry over to analysis of the edge-trace record.

For the chart reading versus scanned distance, refer to Figure 1. The following procedure is employed in choosing points.

1) Determine the end-points first; these are the points where the slope of the trace goes to zero. Scrutinize the trace carefully; the decision is critical. The end points should be located exactly where the slope goes to zero, within the restraints of the one-half scale division previously cited.

2) Select an arbitrary origin, away, from the curve on the lower-density portion of the trace. Measure everything from this point. Since the chart paper usually has divisions of 1/10-inch, it is convenient to use the paper itself for measurement. The listing of the points, in Figure 1 illustrates this.

3) Choose as many points along the curve as is consistent with its delineation, listing them (both chart reading and scanned distance) to the nearest one-half scale division. Figure 1 illustrates the points chosen and lists all those used. Use as many as needed to characterize the trace accurately: but do not use less than 18 not more than 38*. Do not take points so close together that, because of rounding to the nearest half-scale division, there are consecutive pairs having the same value of Chart Reading (e.g., ...28.5,0.20,28.5,0.25,...). This will lead to serious computational problems in the subsequent acutance determination.

4) So that the interpolating routine in the program has sufficient information at the ends of the scale, include points at either end one chart division away from the end point, repeating the value of chart reading. These are placed in lines 917 and 925 of the ACUMTF program listing for emphasis, and must be included in the total number of point-pairs on line 916. Table I shows the listing that would finally accompany the data taken from Figure 1.

5) Determine the trace constant for the microdensitometer; this constant converts inches on the chart to micrometres on the sample, and is a function of scan speed and chart speed. The constant can be found in the microdensitometer operating manual for the various speeds used by the instrument. For the trace shown in the Figure 1, that constant has a value of 3.048.

6) Enter edge-data in ACUMTF as follows:

   a) Line 916 contains the number of pairs of points (chart reading, scan distance) and the trace constant; Thus . . . .916 Data 21,3.048 . . . for the data of Table I.

   b) Line 917 and subsequent contains the points (chart reading, distance), in that order, in pairs, in ascending (increasing) order.

*This is limited by the program array size, lines 015 and 016 of the listing.



EDGE POINTS

Chart Reading	Distance (inches)
15.0	0.30
15.5	0.65
16.0	0.80
17.0	0.95
19.0	1.10
20.5	1.20
22.5	1.30
25.0	1.40
28.0	1.50
31.5	1.60
35.5	1.70
40.0	1.80
45.5	1.90
50.5	2.00
53.0	2.10
56.5	2.25
57.0	2.30
58.0	2.45
58.5	2.70

Figure 1: Typical microdensitometer trace of an edge, showing the particular points chosen for this edge. All data are listed to the nearest one-half scale division, and are chosen (when possible) to be exact.

TABLE I

Complete List of Point-Pairs for Edge-Trace in Figure 1

Chart Reading	Distance (inches)
15.0	0.20
15.0	0.30
15.5	0.65
16.0	0.80
17.0	0.95
19.0	1.10
20.5	1.20
22.5	1.30
25.0	1.40
28.0	1.50
31.5	1.60
35.5	1.70
40.0	1.80
45.5	1.90
50.5	2.00
53.0	2.10
56.5	2.25
57.0	2.30
58.0	2.45
58.5	2.70
58.5	2.80

Consult lines 916 through 925 of the ACUMTF listing for a specific example.

Program Synopsis

Once the data have been patched-in, the program carries out the following operations in sequence: it first reads the data into the program and prints it. This allows a visual check of the calibration information. The program then calculates a calibration look-up table, but does not print it. A spline-fitting technique is used to accomplish this [2].

The edge-trace data are now read into the program, converting the chart readings to density, through the look-up table previously prepared, and the scanned distance from inches to micrometres. The density values are converted to transmittance and the data are inverted (in order of increasing distance and transmittance) in anticipation of eventual use in determining transfer function. The basic edge-trace information is then printed out, together with the adjusted edge-transmittance versus distance pairings.

The acutance calculation follows this. First the program finds the end points by searching for the points where $(\Delta D/\Delta x) = 0.005$. Since the data are digitized, there may not be a point that has that exact value in the listing. Therefore, the program chooses the next higher value at the low-density end of the trace and the next lower value at the high-density end of the trace. To provide the user with the value of the point that preceded the one used in the calculation, it is printed to the right of the end-points in the listing of Figure 2. This manner of choice is entirely arbitrary, and may lead to a value of acutance that is slightly higher than the "true" value. Once the end-points have been chosen, the values of the slope are squared and summed. Acutance is then calculated. The values of D_b and D_a are printed out, respectively as D-MAX and D-MIN, with the values of the slope at the end-points following (and identified). The value of acutance, rounded to three significant figures, is then printed. For a complete print-out of a typical ACUMTF run, see Figure 2.

At this point, the user is given a choice. If he is interested in continuing the calculations to transfer function, it is necessary to transfer the pertinent edge-data (in terms of transmittance and distance) to another program. ACUMTF prints the instructions for this step, and asks the user to respond with a "1" if transfer function calculation are required, and a "2" if not. Both printouts and responses are shown in Appendix A, following the program listing. It should be noted that the two outermost data points, one on either side of the edge-data corresponding to the list in Table I, are not put into the data file. Subsequently, the program notifies the user of the data transfer and instructs him to run the program EMTFER. No other input is required for that program to be executed. However, it may not be executed without first running ACUMTF.

BASIC CALIBRATION DATA

CHART READING	DIFFUSE DENSITY
.0	.00
1.0	.04
10.5	.35
17.0	.60
25.5	.88
41.0	1.42
54.5	1.87
65.0	2.26
72.5	2.55
79.0	2.78
91.5	3.20

BASIC EDGE-TRACE DATA

CHART READING	DIFFUSE DENSITY	SAMPLE DISTANCE	ADJUSTED EDGE TRANSMITTANCE	
15.0	.52	.610	.000	.009700
15.0	.52	.914	.305	.009700
15.5	.54	1.981	1.067	.010118
16.0	.56	2.438	1.524	.011001
17.0	.60	2.896	1.676	.011465
19.0	.67	3.353	2.134	.015190
20.5	.72	3.658	2.438	.018424
22.5	.79	3.962	2.743	.026887
25.0	.86	4.267	3.048	.041163
28.0	.96	4.572	3.353	.059385
31.5	1.08	4.877	3.658	.082447
35.5	1.23	5.182	3.962	.108994
40.0	1.39	5.486	4.267	.136772
45.5	1.57	5.791	4.572	.163902
50.5	1.73	6.096	4.877	.189787
53.0	1.82	6.401	5.182	.212938
56.5	1.94	6.858	5.639	.251189
57.0	1.96	7.010	6.096	.274554
58.0	1.99	7.468	6.553	.287400
58.5	2.01	8.230	7.620	.301012
58.5	2.01	8.534	7.925	.301012

D-MAX = 2.01 HIGH-END SLØPE = .0050 .0041
D-MIN = .52 LØW-END SLØPE = .0050 .0043

ACUTANCE = 51600

Figure 2: Typical printout of ACUMTF.

Error Analysis:

The possible error in the calculation of acutance is related directly to the specification that data be read from the chart to the nearest 1/2-scale division. In the ordinate (density), the microdensitometer is adjusted to cover from 0 to 3.0 over 80 full scale divisions. This gives a factor of 0.02 density units/half-scale division. We will assume that the careful operator will make no more than a one-quarter scale division error in reading, so that the maximum error in density should be no more than 0.01.

In the abscissa (scan distance), we will use the slowest microdensitometer speed. This provides 3.048 micrometres on the sample per inch on the chart. The chart is marked in 1/10-inch scale divisions. With this, on the sample, we will have 0.15 micrometres/half-scale division. Assuming the same quarter-scale division reading error that we did for density, the maximum error in scan distance (on the sample) is estimated to be no more than 0.075 micrometres.

The program spline-fits and expands the calibration and edge-data, the latter to approximately 350 points. The basis of this fit is a 3rd-degree polynomial and the transitions are smooth (the derivatives must match at the ends of adjoining fitted segments). We will therefore assume there is no significant contribution to the error in acutance from the program except for that associated with the choice of end-points previously discussed. Because density and chart reading can only be specified to three significant figures, information derived from the calculations of the programs considered in this note will be printed to no greater than three significant figures.

Referring to the input data (taken from the edge-trace record and the density calibration data), we can show that the estimated relative error in acutance is given approximately by

$$E(rel) = \frac{4}{N}\left[\frac{\beta}{\overline{\Delta D}} + \frac{\xi}{\overline{\Delta x}}\right] + \frac{2\beta}{D_b - D_a},$$

where N = number of data points;

β = measurement error in density (0.01);

ξ = measurement error in scan distance (0.075 μm);

$\overline{\Delta D}$ = average tabular density difference;

$\overline{\Delta x}$ = average tabular scan distance difference (μm);

D_b = maximum density for acutance calculation;

D_a = minimum density for acutance calculation.

For the data shown in Figure 2,

$$N = 19;$$

$$\overline{\Delta D} = 0.08;$$

$$\overline{\Delta x} = 0.41;$$

$$D_b = 2.01;$$

$$D_a = 0.52.$$

When these are inserted into the above equation, we obtain

$$E(rel) = 8\%,$$

and the value of acutance is properly given as

$$\text{Acutance} = 51{,}600 \pm 4100.$$

These estimates of accuracy are consistent with this method of data analysis in the time-sharing computer. The calculated error can be reduced somewhat by using more data points, but to make a significant improvement in accuracy, it will be necessary to take data from the microdensitometer on magnetic tape and subsequently process the information in a more comprehensive program. It should also be noted that the expression for relative error is only approximate: the error changes inversely as a function of acutance. As calculated above, it represents maximum relative error. Since the data for subsequent programs documented in this note depend on this calculation, the same accuracy can be ascribed to their results. As a general guide, based on these considerations and experience with the procedure at NBS, the relative error in acutance (for the techniques outlined here) can be taken to be not less than 5% nor greater than 10%.

EMTFER

The program EMTFER calculates the optical transfer function of the microdensitometer from analysis of the edge-trace record, on the assumption that the edge quality is sufficiently high to warrant no corresponding correction to the trace-data*. When the edge is moved through the optical axis of the instrument (assuming a slit has been adjusted parallel to the edge and that the adjustment of all optics and related mechanical fixtures has been optimized), the illuminance falling on the photodetector will be given by

$$I(x) = \int E(x - x')f(x')\,dx', \tag{2}$$

where E(x') is the edge-object radiance and f(x') is the impulse response of the optical system. In practice, the photometer output is converted to chart reading. In turn, this is converted to density and then transmittance (the equivalent of illuminance in this context) through the preceding program, ACUMTF. To simplify the presentation, we will retain the identity of the edge through use of E(x') and bypass the chart reading-density-transmittance procedure.

Because the edge is the signum function, i.e.,

$$\begin{aligned} E(x') &= \mathrm{sgn}(x'), \\ &= 1; \qquad (0 < x') \\ &= -1; \qquad (0 > x') \end{aligned} \tag{3}$$

we can rewrite Eq. (2) as

$$I(x) = \int_{-\infty}^{x} f(x')\,dx'. \tag{4}$$

*When the edge is of lesser quality, the result is an admixture of the MTF and the edge characteristics, and what is determined is its spectrum. On the other hand, for a given edge-making process, a measure of the consistency of the process and its analysis through ACUMTF and EMTFER is the accuracy with which this "MTF" is reproduced from edge to edge for the same set of optical parameters. It is also a useful means for checking instrument adjustment; on the same edge, the "MTF" should be reproduced. If not, the instrument will have been incorrectly adjusted.

When we differentiate both sides of this equation, we obtain

$$f(x) = \frac{d}{dx} \Big(I(x) \Big), \tag{5}$$

and we have thus derived the impulse response of the optical system. Since the Fourier Transform of the impulse response is the transfer function, we will have

$$\tilde{f}(\sigma) = \int f(x) \exp(-2\pi i \sigma x)\, dx$$

$$= \int \left[\frac{d}{dx} I(x)\right] \exp(-2\pi i \sigma x)\, dx, \tag{6}$$

and the Fourier Transform of the differentiated edge-image distribution is the optical transfer function.

This is clearly an idealization. For relatively noise-free edges such as those produced at NBS, no allowances need be made for the noise contributed by the grain in the emulsion of the photographic process, i.e., no smoothing of the edge-data is required. On the other hand, when edge-gradient analysis is used to obtain MTF from noisy data, large errors can result unless the data are smoothed (filtered): there can be little confidence in the results. Because of this, edge-gradient analysis is (generally) not a particularly accurate technique. Even allowing for a certain amount of smoothing, the procedure suffers from the fact that because the spectrum of an edge falls off as the reciprocal of spatial frequency, the data in those regions of particular interest (the high-frequency portion of the curve) are produced with a low signal-to-noise ratio. There are still other considerations that mitigate against the accuracy and precision of this method. However, the technique is easily implemented and results can be interpreted with little ambiguity. In many cases, it is the only available technique. But the user should not assign undue confidence to the results; 10% RMS error in transfer function is often about the best that can be achieved; with experience and increased technical skill in carrying out the measurements, this can sometimes be reduced to 5%. Nothing better than this is warranted and should not be expected.

The program EMTFER embodies the operations described by Eqs. (2) through (6). Because the differentiation of the edge produces the impulse response with values of zero on either side, a direct Fourier transformation can be carried out. For reasons of accuracy, the numerical integration was implemented through Weddle's Rule [3].

Weddle's Rule requires that there be N = 6M + 1 data points in the calculation, where M is an integer chosen large enough for accuracy and small enough to keep the available arrays within bounds. In keeping with the limitations of the size of BASIC programs, a value of 33 was chosen for M, which thereby gave 199 for N, and an array of 200 values is used for the edge analyzed in the program. The given edge information is expanded by interpolation through a spline-fitting technique [2] and occupies the full 200-element array. Subsequently, because the spline-fit coefficients are available, the derivative is calculated on the basis of the algorithm.

Data Input to EMTFER:

The input for EMTFER comes from ACUMTF. It is left in a file within the computer, from which EMTFER draws. it. EMTFER cannot be executed unless ACUMTF is first run. A complete listing of EMTFER is shown in Appendix B.

Program Synopsis:

Once the data file has been provided, the program carries out the following operations. Before processing the data, the program asks the user if a plot of the MTF is required. If the response is affirmative, a data file is set up immediately and filled during the ensuing calculations. This is a necessary because EMTFER uses nearly all of the available space in the computer and another array cannot be set up.

The program then reads the data and prints it (a patch-routine is subsequently provided that permits skipping the printing: see Appendix B, following the listing). This allows a check on the input data (it should agree with the edge-list printed by ACUMTF). The edge data are next expanded into the 200-element array and the derivative calculated. Following this, the array is set up for integration by Weddle's Rule and the array summed: this is the area under the impulse response and it is used for normalizing the transfer function.

The Fourier transform of the edge distribution is next calculated. The cosine and sine transforms are calculated separately so that the modulus and phase of the optical transfer function can be determined. As the calculation is made, the results are printed, frequency by frequency, and the data file is filled. The print-out is based on increments of 10 cycles/mm. this value is arbitrary, and is contained in line 185 of the listing;

```
185 F = 10 .
```

It can be changed by patching-in the desired value prior to execution of the program*.

The print-out (and calculation) is stopped arbitrarily at an MTF value of 0.04. This threshold value can be changed if desired. The appropriate line in EMTFER reads:

216 IF (S3 - 0.04),302,302,220 .

If a higher or lower threshold is required, re-write the line with the desired value substituted for 0.04.

When a plot is required (and has been indicated prior to carrying out the calculations), a data file will have been set up for the program MTFPLT and after the completion of the transfer function calculations the computer will notify the user that MTFPLT should next be run. Typical print-out of these instructions and their responses are shown in Appendix B, following the program listing.

The print-out of EMTFER is in two parts. The first lists the edge information, following the question/answer about plotting. This is shown is Figure 3. The second is the listing of the MTF, and is shown in Figure 4.

MTFPLT

The program MTFPLT plots the modulation transfer function (MTF) calculated by the program EMTFER through edge-gradient analysis. It is specifically tailored for this single purpose, and assumes the 10 cycle/mm increment in spatial frequency used by EMTFER in its calculations. MTFPLT derives its input data only from EMTFER.

Program Synopsis:

The program reads the data from the file set up by EMTFER. The fundamental plotting increment along the ordinate scale is 0.04. The data are then tested against this increment, sorted accordingly, and

*The user is cautioned that the routine for plotting the MTF assumes this value of spatial frequency as an increment. MTFPLT must be modified accordingly, if line 185 is changed and a plot is subsequently desired.

```
?BASIC EMTFER
RUN
IF A PLØT ØF THE MTF IS DESIRED, TYPE 1 AFTER !
ØTHERWISE, TYPE 2.
!1
```

BASIC EDGE-DATA

TRACE DISTANCE (MICRØMETERS)	TRANSMITTANCE
.000	.009700
.762	.010118
1.219	.011001
1.372	.011465
1.829	.015190
2.134	.018424
2.438	.026887
2.743	.041163
3.048	.059385
3.353	.082447
3.658	.108994
3.962	.136772
4.267	.163902
4.572	.189787
4.877	.212938
5.334	.251189
5.791	.274554
6.248	.287400
7.315	.301012

Figure 3: Typical edge-information printout of EMTFER, showing the initial program instruction for subsequent plotting of MTF.

TRANSFER FUNCTION

FREQ.	MTF	PHASE
0	1.000	.000
10	.997	.264
20	.989	.527
30	.976	.791
40	.957	1.054
50	.934	1.318
60	.906	-1.560
70	.874	-1.297
80	.839	-1.034
90	.800	-.771
100	.759	-.509
110	.715	-.247
120	.670	.015
130	.624	.276
140	.577	.536
150	.531	.796
160	.485	1.054
170	.439	1.312
180	.396	1.568
190	.353	-1.320
200	.313	-1.068
210	.275	-.819
220	.240	-.573
230	.206	-.331
240	.176	-.094
250	.148	.135
260	.123	.357
270	.101	.566
280	.081	.760
290	.063	.929
300	.049	1.062
310	.036	1.135

Figure 4: Typical printout of EMTFER.

rounded to the nearest 0.04. The graph axes and points are then printed together, starting with the topmost ordinate value and working down to the x-axis. Increments on the x-axis are 10 cycles/mm.

The program is presently limited to positive values of MTF; only one curve can be plotted. Multiple-valued functions (such as those produced by a central obstruction in the aperture) will not plot correctly with the present sorting algorithm.

A typical print-out from this program is shown in Figure 5.



Figure 5: Typical printout of MTFPLT.

REFERENCE:

[1] C.E.K. Mees and T.H. James, The Theory of the Photographic Process, 3rd Ed. (New York, The MacMillan Company, 1966), p. 511.

[2] R. H. Pennington, Introductory Computer Methods and Numerical Analysis, 2nd Ed. (The MacMillan Company, Collier MacMillan Canada, Ltd, Toronto, Ontario, 1970), pp. 445-454.

[3] J. B. Scarborough, Numerical Mathematical Analysis, 2nd Ed. (The Johns Hopkins Press, Baltimore, Md., 1950), pp. 133-137, 180.

APPENDIX A

ACUMTF - A Program for Calculation of Acutance

```
001 * RICHARD E. SWING, 232.08, X2159, PRØGRAM: ACUMTF
002 *
003 * THIS PRØGRAM CALCULATES ACUTANCE FRØM THE TRACE ØF AN
004 * EDGE BY A MICRØDENSITØMETER AND PREPARES DATA FØR ITS
005 * EVENTUAL USE IN CALCULATING TRANSFER FUNCTIØN THRØUGH
006 * THE PRØGRAM CALLED EMTFER.
007 *
015 DIM X(40),Y(40),D(40),P(40),E(40),C(4,40),A(40,4)
016 DIM B(40),Z(40),K(40),F(40),T(40),V(40)
017 DIM G(350),H(350),L(350),R(350)
018 *
020 Q1 = 1
021 FØR J = 1 TØ 5
022 PRINT
023 NEXT J
024 READ M
025 GØ TØ 270
026 FØR I = 1 TØ M          'INPUT DENSITY CALIBRATIØN DATA'
027 READ X(I),Y(I)
028 PRINT,029,X(I),Y(I)
029 FMT X21,F8.1,X4,F8.2
030 NEXT I
032 M1 = M-1               'CALCULATE FIT-CØEFFICIENTS'
034 FØR K=1 TØ M1
036 D(K) = X(K+1)-X(K)
038 P(K) = D(K)/6
040 E(K) = (Y(K+1)-Y(K))/D(K)
042 NEXT K
044 FØR K = 2 TØ M1
046 B(K) = E(K)-E(K-1)
048 NEXT K
050 A(1,2) = -1.0 - D(1)/D(2)
052 A(1,3) = D(1)/D(2)
054 A(2,3) = P(2)-P(1)*A(1,3)
056 A(2,2) = 2.0*(P(1)+P(2))-P(1)*A(1,2)
058 A(2,3) = A(2,3)/A(2,2)
060 B(2) = B(2)/A(2,2)
062 FØR K = 3 TØ M1
064 A(K,2) = 2*(P(K-1)+P(K))-P(K-1)*A(K-1,3)
066 B(K) = B(K)-P(K-1)*B(K-1)
068 A(K,3) = P(K)/A(K,2)
070 B(K) = B(K)/A(K,2)
072 NEXT K
074 Q = D(M-2)/D(M-1)
076 A(M,1) = 1 + Q + A(M-2,3)
078 A(M,2) = -Q-A(M,1)*A(M-1,3)
080 B(M) = B(M-2)-A(M,1)*B(M-1)
082 Z(M) = B(M)/A(M,2)
084 M2 = M-2
086 FØR I = 1 TØ M2
088 K = M - I
090 Z(K) = B(K)-A(K,3)*Z(K+1)
092 NEXT I
094 Z(1) = -A(1,2)*Z(2) - A(1,3)*Z(3)
```

```
096 FØR K = 1 TØ M1
098 Q = 1/(6*D(K))
100 C(1,K) = Z(K)*Q
102 C(2,K) = Z(K+1)*Q
104 C(3,K) = (Y(K)/D(K))-(Z(K)*P(K))
106 C(4,K) = (Y(K+1)/D(K))-(Z(K+1)*P(K))
108 NEXT K
109 IF Q1 = 2 GØ TØ 222
110 B = X(M)/(0.50)      'GENERATE INDEPENDENT VARIABLE VALUES'
114 FØR J = 1 TØ B
115 IF Q1 = 2 GØ TØ 118
116 G(J) = X(1) + (J-1)*(0.50)
117 GØ TØ 122
118 G(J) = (J-1)*(1/M7)
122 IF (G(J)-X(1)),146,124,128    'GENERATE DEPENDENT VARIABLE'
124 H(J) = Y(1)
126 GØ TØ 147
128 K = 1
130 IF (G(J)-X(K+1)),140,132,136
132 H(J) = Y(K+1)
134 GØ TØ 147
136 K = K+1
138 IF (G(J)-X(M)),130,146,146
140 H(J) = (X(K+1)-G(J))*(C(1,K)*((X(K+1)-G(J))**2)+C(3,K))
142 H(J) = H(J)+(G(J)-X(K))*(C(2,K)*((G(J)-X(K))**2)+C(4,K))
143 GØ TØ 147
146 H(J) = Y(M)
147 NEXT J
148 IF Q1=2 GØ TØ 230
149 GØ TØ 290
150 PRINT
154 PRINT,156
156 FMT X8,"CHART",X6,"DIFFUSE",X7,"SAMPLE",X9,"ADJUSTED EDGE"
158 PRINT,160
160 FMT X7,"READING",X5,"DENSITY",X6,"DISTANCE",X8,"TRANSMITTANCE"
161 PRINT
162 READ M2,N
163 W1 = M2
166 FØR J = 1 TØ M2
168 READ K(J),F(J)     'INPUT EDGE-TRACE DATA'
170 IF (K(J)-G(1)),172,172,176    'CØNVERT TØ DENSITY'
172 L(J) = H(1)
174 GØ TØ 188
176 K = 1
178 IF (K(J)-G(K+1)),180,180,184
180 L(J) = H(K+1)
182 GØ TØ 188
184 IF (K(J)-G(K+1)) < 0.00001 THEN 180
185 K = K+1
186 GØ TØ 178
188 NEXT J
189 FØR J = 1 TØ M2
190 T(J) = 10**(-L(M2+1-J))
```

```
192 V(J) = N*(F(M2)-F(M2+1-J))
194 PRINT,200,K(J),L(J),N*F(J),V(J),T(J)
200 FMT X5,F8.1,X4,F8.2,X5,F8.3,X5,F8.3,X3,F8.6
202 NEXT J
210 Q1 = 2
214 FØR J = 1 TØ M2
216 X(J) = N*(F(J)-F(1))
218 Y(J) = L(J)
219 NEXT J
220 M = M2
221 GØ TØ 032
222 M6 = INT(1+X(M))
223 M7 = INT(350/M6)
224 B = M6*M7
228 GØ TØ 114
230 FØR K = 1 TØ (B-1)
232 R(K) = (M7)*(H(K+1)-H(K))   'CØMPUTE (DELTA-D)/(DELTA-X)'
234 NEXT K
236 K = 1
238 IF (R((B/4)-K)-0.0050),239,241,244  'LØW-END CUTØFF TEST'
239 M4 = (B/4)-K
240 GØ TØ 248
241 M4 = (B/4)-K-1
242 GØ TØ 248
244 K = K+1
246 GØ TØ 238
248 K = 1
250 IF (R(B-(B/4)+K)-0.0050),251,253,256  'HIGH-END CUTØFF TEST'
251 M5 = (B/4)-K+1
252 GØ TØ 260
253 M5 = (B/4)-K
254 GØ TØ 260
256 K = K+1
258 GØ TØ 250
260 P9 = H(B-M5)-H(M4)
266 S = 0
268 GØ TØ 296
270 PRINT "                          BASIC CALIBRATIØN DATA"
272 PRINT
274 PRINT,276
276 FMT X24,"CHART",X6,"DIFFUSE"
278 PRINT,280
280 FMT X23,"READING",X5,"DENSITY"
286 PRINT
288 GØ TØ 026
290 PRINT
291 PRINT
292 PRINT "                          BASIC EDGE-TRACE DATA"
294 GØ TØ 150
296 FØR J = M4 TØ (B-M5)
298 S = S + (R(J))↑2       'SUM ØF SQUARED-SLØPE VALUES'
300 NEXT J
```

```
301 G2 = (10↑6)*S/(B-M5-M4)
302 A = G2/P9                          'CALCULATE ACUTANCE'
303 GØ TØ 310
304 PRINT,305,A
305 FMT "ACUTANCE = ",I7
306 FØR J = 1 TØ 15
307 PRINT
308 NEXT J
309 GØ TØ 325
310 PRINT
312 PRINT,314,H(B-M5),R(B-M5),R(B-M5+1)
314 FMT "D-MAX = ",F8.2,X4,"HIGH-END SLØPE = ",F8.4,X8,F8.4
316 PRINT,318,H(M4),R(M4),R(M4-1)
318 FMT "D-MIN = ",F8.2,X4,"LØW-END SLØPE  = ",F8.4,X8,F8.4
320 PRINT
322 GØ TØ 370
325 PRINT
326 PRINT
327 PRINT "IF MTF CALCULATIØNS ARE TØ FØLLØW, TYPE 1 AFTER !."
328 PRINT "ØTHERWISE, TYPE 2, AND THE RUN WILL TERMINATE."
331 INPUT A
333 IF A = 1 GØ TØ 338
334 IF A = 2 GØ TØ 999
336 PRINT
337 GØ TØ 325
338 FILES,DATAC
339 FØR J = 2 TØ W1-1
340 WRITE #2,(V(J)-V(2)),T(J)
341 NEXT J
342 PRINT
343 PRINT "AN EDGE-DATA FILE HAS BEEN SET UP; NØW RUN THE"
344 PRINT "PRØGRAM EMTFER TØ ØBTAIN THE TRANSFER FUNCTIØN."
346 PRINT
347 GØ TØ 999
370 IF A => 100000 GØ TØ 382
371 IF A => 10000 GØ TØ 378
374 A = INT((A+5.0)/10)*10
376 GØ TØ 384
378 A = INT((A+50.0)/100)*100
380 GØ TØ 384
382 A = INT ((A+500.0)/1000)*1000
384 GØ TØ 304
899 *                        DATA SECTIØN
900 DATA 11
910 DATA 0,0,1.0,0.04,10.5,0.345,17.0,0.60,25.5,0.88
912 DATA 41.0,1.42,54.5,1.87,65.0,2.26
914 DATA 72.5,2.55,79.0,2.78,91.5,3.20
916 DATA 21,3.048
917 DATA 15.0,0.20
918 DATA 15.0,0.30,15.5,0.65,16.0,0.80,17.0,0.95,19.0,1.10
920 DATA 20.5,1.20,22.5,1.30,25.0,1.40,28.0,1.50,31.5,1.60
922 DATA 35.5,1.70,40.0,1.80,45.5,1.90,50.5,2.00,53.0,2.10
924 DATA 56.5,2.25,57.0,2.30,58.0,2.45,58.5,2.70
925 DATA 58.5,2.80
999 END
```

At the end of the Acutance calculation, the program spaces enough lines to provide paper for trimming to an 11-inch length and then asks the user to choose between two options. One of these is to proceed to the calculation of Transfer Function. The following is a printout of this, together with the affirmative response:

```
IF MTF CALCULATIØNS ARE TØ FØLLØW, TYPE 1 AFTER !.
ØTHERWISE, TYPE 2, AND THE RUN WILL TERMINATE.
!1

AN EDGE-DATA FILE HAS BEEN SET UP; NØW RUN THE
PRØGRAM EMTFER TØ ØBTAIN THE TRANSFER FUNCTIØN.


  999 EXIT

?BASIC EMTFER
```

The second option is the termination of the run without further calculation. The following is a printout of the alternate response to the program options.

```
IF MTF CALCULATIØNS ARE TØ FØLLØW, TYPE 1 AFTER !.
ØTHERWISE, TYPE 2, AND THE RUN WILL TERMINATE.
!2

  999 EXIT

?
```

ACUMTF has been designed to be used at NBS for calibration of high-quality edges. The rounding algorithm therefore does not extend to values of acutance less than 1000 (lines 370 through 384). For lower quality edges, this program should not be used. As a means of assessing this, examination of the listing of sample distance in the ADJUSTED EDGE TRANSMITTANCE column of Figure 2 will provide a useful criterion. If the final distance value is less than 15, the edge may be considered of sufficiently high quality to warrant further use of the program. A value larger than this will require program modifications which will not be discussed here.

APPENDIX B

EMTFER - A Program for Calculation of Transfer Function

```
001 * RICHARD E. SWING,232.08,X2159,PRØGRAM: EMTFER
002 *
003 * THIS PRØGRAM CALCULATES TRANSFER FUNCTIØN FRØM AN EDGE-
004 * TRACE, THE DATA FØR WHICH IS DERIVED FRØM THE PRØGRAM
005 * CALLED ACUMTF.  EMTFER CANNØT BE EXECUTED WITHØUT FIRST
006 * RUNNING ACUMTF.  EMTFER ALSØ SUPPLIES THE NECESSARY
007 * DATA TØ THE PRØGRAM MTFPLT FØR THE PLØTTING ØF MTF.
008 *
011 DIM F(200),L(200)
012 DIM X(40),Y(40),D(40),P(40),E(40),C(4,40),A(40,4)
013 DIM B(40),Z(40),G(200),H(200)
015 *
016 PRINT
017 Q1 = 1
018 GØ TØ 308
019 GØ TØ 266
020 FILES,DATAC,DATAB
021 K = 0
022 READ #2,X(K+1),Y(K+1)
023 IF END #2 THEN 030
026 PRINT,027,X(K+1),Y(K+1)
027 FMT X15,F8.3,X11,F8.6
028 K = K + 1
029 GØ TØ 022
030 M = K
031 PRINT
032 M1 = M-1                'CALCULATE FIT-CØEFFICIENTS'
034 FØR K=1 TØ M1
036 D(K) = X(K+1)-X(K)
038 P(K) = D(K)/6
040 E(K) = (Y(K+1)-Y(K))/D(K)
042 NEXT K
044 FØR K = 2 TØ M1
046 B(K) = E(K)-E(K-1)
048 NEXT K
050 A(1,2) = -1.0 - D(1)/D(2)
052 A(1,3) = D(1)/D(2)
054 A(2,3) = P(2)-P(1)*A(1,3)
056 A(2,2) = 2.0*(P(1)+P(2))-P(1)*A(1,2)
058 A(2,3) = A(2,3)/A(2,2)
060 B(2) = B(2)/A(2,2)
062 FØR K = 3 TØ M1
064 A(K,2) = 2*(P(K-1)+P(K))-P(K-1)*A(K-1,3)
066 B(K) = B(K)-P(K-1)*B(K-1)
068 A(K,3) = P(K)/A(K,2)
070 B(K) = B(K)/A(K,2)
072 NEXT K
074 Q = D(M-2)/D(M-1)
076 A(M,1) = 1 + Q + A(M-2,3)
078 A(M,2) = -Q-A(M,1)*A(M-1,3)
080 B(M) = B(M-2)-A(M,1)*B(M-1)
082 Z(M) = B(M)/A(M,2)
084 M2 = M-2
```

```
086 FØR I = 1 TØ M2
088 K = M - I
090 Z(K) = B(K)-A(K,3)*Z(K+1)
092 NEXT I
094 Z(1) = -A(1,2)*Z(2) - A(1,3)*Z(3)
096 FØR K = 1 TØ M1
098 Q = 1/(6*D(K))
100 C(1,K) = Z(K)*Q
102 C(2,K) = Z(K+1)*Q
104 C(3,K) = (Y(K)/D(K))-(Z(K)*P(K))
106 C(4,K) = (Y(K+1)/D(K))-(Z(K+1)*P(K))
108 NEXT K
110 B = 200          'CALCULATE EDGE-DERIVATIVE'
112 FØR J = 1 TØ B
114 G(J) = (J-1)*(X(M)-X(1))/B
116 IF (G(J)-X(1)),118,118,122
118 H(J) = 0
120 GØ TØ 136
122 K = 1
124 IF (G(J)-X(K+1)),126,126,132
126 H(J) = -3*C(1,K)*(X(K+1)-G(J))**2+3*C(2,K)*(G(J)-X(K))**2
128 H(J) = H(J)-C(3,K)+C(4,K)
130 GØ TØ 136
132 K = K+1
134 IF (G(J)-X(M)),124,118,118
136 NEXT J
140 FØR J = 1,199,2  'SET UP ARRAY FØR WEDDLE-INTEGRATIØN'
142 H(J) = H(J)*1
144 NEXT J
146 FØR J = 7,193,6
148 H(J) = H(J)*2
150 NEXT J
152 FØR J = 2,194,6
154 H(J) = H(J)*5
156 NEXT J
158 FØR J = 6,198,6
160 H(J) = H(J)*5
162 NEXT J
164 FØR J = 4,196,6
166 H(J) = H(J)*6
168 NEXT J
170 S = 0
171 D = (0.30)*(X(M)-X(1))/B
172 FØR J = 1 TØ 199
174 S = S+H(J)*D  'AREA UNDER SPREAD FUNCTIØN'
176 NEXT J
177 FØR J = 1 TØ 15
178 PRINT
179 NEXT J
180 GØ TØ 290
182 K = 0
184 FØR J = 1 TØ 199
185 F = 10
186 F(J) = CØS((6.28318)*K*F*(0.001)*G(J))
188 L(J) = SIN((6.28318)*K*F*(0.001)*G(J))
```

```
190 F(J) = F(J)*H(J)
192 L(J) = L(J)*H(J)
194 NEXT J
196 S1 = 0
198 S2 = 0
200 FØR J = 1 TØ 199
202 S1 = S1+F(J)*D        'CØSINE TRANSFØRM'
204 S2 = S2+L(J)*D        'SINE TRANSFØRM'
206 NEXT J
208 S3 = (1/S)*SQR((S1)↑2 + (S2)↑2)  'MØDULUS (MTF)'
209 S4 = ATN((S2)/(S1))              'PHASE (PTF) '
210 PRINT,211,K*F,S3,S4
211 FMT X17,I4,X6,F8.3,X3,F8.3
212 IF Q1 = 2 GØ TØ 216
213 WRITE #3,S3
216 IF (S3-0.04),302,302,217
217 IF Q1 = 2 GØ TØ 220
218 IF K*10 = 500 GØ TØ 302
220 K = K+1
222 GØ TØ 184
266 FØR J = 1 TØ 15
268 PRINT
269 NEXT J
270 PRINT "                          BASIC EDGE-DATA"
272 PRINT
274 PRINT,276
276 FMT X15,"TRACE DISTANCE",X4,"TRANSMITTANCE"
278 PRINT,280
280 FMT X15,"(MICRØMETERS)"
286 PRINT
288 GØ TØ 020
290 PRINT
292 PRINT
293 PRINT "                            TRANSFER FUNCTIØN"
294 PRINT
296 PRINT,298
298 FMT X18,"FREQ.",X7,"MTF",X7,"PHASE"
299 PRINT
300 GØ TØ 182
302 FØR J = 1 TØ 15
304 PRINT
306 NEXT J
307 GØ TØ 318
308 PRINT "IF A PLØT ØF THE MTF IS DESIRED, TYPE 1 AFTER !"
310 PRINT "ØTHERWISE, TYPE 2."
311 INPUT C
312 IF C = 1 GØ TØ 019
313 IF C = 2 GØ TØ 315
314 GØ TØ 308
315 Q1 = 2
316 GØ TØ 019
318 IF Q1 = 2 GØ TØ 999
319 PRINT "AN MTF DATA FILE HAS BEEN SET UP; NØW RUN THE"
320 PRINT "PRØGRAM MTFPLT TØ ØBTAIN A PLØT."
322 PRINT
324 PRINT
999 END
```

Prior to the transfer function calculation, the program asks the user to decide about a plot of the MTF:

```
IF A PLØT ØF THE MTF IS DESIRED, TYPE 1 AFTER !
ØTHERWISE, TYPE 2.
```

Should the user respond with a "2", the calculation will be terminated after the printing of the transfer function. When the response is a "1", a data file is, prepared during calculation for use with the subsequent plotting program. The user is notified when the file is ready by the following statement:

```
AN MTF DATA FILE HAS BEEN SET UP; NØW RUN THE
PRØGRAM MTFPLT TØ ØBTAIN A PLØT.
```

```
999 EXIT
```

The user now executes the program MTFPLT to obtain the plot.

When EMTFER obtains the edge information from the file set up by ACUMTF, it prints-out a list. This can then be checked with the listing in the print-out of ACUMTF. It is not always necessary to list this information and there is a patch-routine to suppress the printing. It is as follows:

```
?PATCH EMTFER

BEGIN
026
027
266 PRINT
268
269
270
272
274
276
278
280
286

?BASIC EMTFER
```

APPENDIX C

MTFPLT - A Program for Plotting Transfer Function

```
001 * RICHARD E. SWING, 232.08, X2159, PRØGRAM: MTFPLT
002 *
003 * THIS PRØGRAM PLØTS THE MØDULATIØN TRANSFER FUNCTIØN
004 * PRØDUCED BY THE PRØGRAM CALLED EMTFER.  IT IS THE THIRD
005 * IN THE SEQUENCE...ACUMTF-EMTFER-MTFPLT....AND CANNØT
006 * BE EXECUTED WITHØUT THE DATA PRØDUCED BY THE ØTHER TWØ
007 * PRØGRAMS.
008 *
016 DIM V(30),A(30),M(50)
018 PRINT
020 FILES,,DATAB
030 K = 0
040 READ #3,M(K)
050 IF END #3 THEN 080
060 K = K+1
070 GØ TØ 040
080 M = K-1
082 PRINT
087 PRINT
090 FØR J = 0 TØ 25
100 V(J) = 0
110 A(J) = 0
120 NEXT J
130 FØR J = 0 TØ M
140 K = 0
150 IF ((25-K)*0.04-M(J)),160,160,190
160 V(K) = 1
170 A(K) = A(K)+1
180 GØ TØ 250
190 IF ((25-K)*0.04-M(J)) < 0.0400 GØ TØ 220
200 K = K+1
210 GØ TØ 150
220 IF ((25-K)*0.04-M(J)) <= 0.0200 GØ TØ 160
230 V(K+1) = 1
235 A(K+1) = A(K+1)+1
250 NEXT J
275 FØR J = 1 TØ 15
280 PRINT
285 NEXT J
290 PRINT,295
295 FMT X24,"MØDULATIØN TRANSFER FUNCTIØN"
300 PRINT
305 PRINT
310 PRINT
311 FØR J = 1 TØ 2
312 PRINT "                   ";
313 PRINT "+"
314 NEXT J
315 FØR J = 0 TØ 24
320 PRINT "        ";
325 IF J = 0 GØ TØ 360
330 IF ((25-J)*4/5) = INT((25-J)*4/5) THEN 350
335 PRINT "           ";
340 PRINT "+";
345 GØ TØ 390
350 PRINT (25-J)*(0.04);
355 GØ TØ 380
```

```
360 PRINT "     ";
365 PRINT "1.0 ";
370 PRINT "*";
375 GØ TØ 390
380 PRINT " ";
385 PRINT "-";
390 IF V(J) = 0 GØ TØ 396
395 GØSUB 600
396 PRINT
400 NEXT J
420 PRINT "           ";
425 FØR J = 1 TØ 55
435 IF (J/5) = INT(J/5) THEN 460
440 PRINT "-";
445 GØ TØ 465
460 PRINT "!";
465 NEXT J
470 FØR J = 0,10,2
475 IF J > 0 GØ TØ 495
480 PRINT "                    ";J;
485 GØ TØ 515
495 PRINT "    ";(50)*J;
500 GØ TØ 515
515 NEXT J
520 PRINT
522 PRINT
525 PRINT,530
530 FMT X24,"SPATIAL FREQUENCY (CYCLES/MM)"
535 PRINT
536 GØ TØ 705
600 S = 0
605 IF J = 0 GØ TØ 685
606 IF J = 1 GØ TØ 632
610 FØR K = 0 TØ J-1
620 S = S+A(K)
630 NEXT K
631 GØ TØ 640
632 S = A(0)
640 FØR K = 0 TØ S-2
650 PRINT " ";
660 NEXT K
670 FØR K = 1 TØ A(J)
680 PRINT "*";
682 NEXT K
683 GØ TØ 700
685 FØR K = 1 TØ A(J)-1
686 PRINT "*";
688 NEXT K
700 RETURN
705 FØR J = 1 TØ 20
710 PRINT
715 NEXT J
999 END
```


USCOMM-NBS-D(

NBS-114A (REV. 7-73)

U.S. DEPT. OF COMM. BIBLIOGRAPHIC DATA SHEET	1. PUBLICATION OR REPORT NO. NBSIR 75-699	2. Gov't Accession No.	3. Recipient's Accession No.
4. TITLE AND SUBTITLE THE CALIBRATION OF PHOTOGRAPHIC EDGES AT NBS			5. Publication Date April 1975
			6. Performing Organization Code
7. AUTHOR(S) Richard E. Swing			8. Performing Organ. Report No.
9. PERFORMING ORGANIZATION NAME AND ADDRESS NATIONAL BUREAU OF STANDARDS DEPARTMENT OF COMMERCE WASHINGTON, D.C. 20234			10. Project/Task/Work Unit No. 2320187
			11. Contract/Grant No.
12. Sponsoring Organization Name and Complete Address (Street, City, State, ZIP) Same as No. 9.			13. Type of Report & Period Covered Interim Report
			14. Sponsoring Agency Code

15. SUPPLEMENTARY NOTES

16. ABSTRACT (*A 200-word or less factual summary of most significant information. If document includes a significant bibliography or literature survey, mention it here.*)

The method by which photographic edges made at NBS are calibrated is presented and discussed in some detail. The programs associated with the computational aspects of the analysis are listed, covered in narrative form, and their limitations and options are presented. The possible use of these edges to determine microdensitometer transfer function is discussed and limitations and relative error of all the calculations and procedures are covered in detail. Program listings are in BASIC language.

17. KEY WORDS (*six to twelve entries; alphabetical order, capitalize only the first letter of the first key word unless a proper name; separated by semicolons*)

Acutance; Calibration; Computer Programs; Microdensitometry; Photographic Edges; Transfer Function.

18. AVAILABILITY [X] Unlimited	19. SECURITY CLASS (THIS REPORT) UNCLASSIFIED	21. NO. OF PAGES 38
[ ] For Official Distribution. Do Not Release to NTIS		
[ ] Order From Sup. of Doc., U.S. Government Printing Office Washington, D.C. 20402, SD Cat. No. C13	20. SECURITY CLASS (THIS PAGE) UNCLASSIFIED	22. Price $3.75
[X] Order From National Technical Information Service (NTIS) Springfield, Virginia 22151		

USCOMM-DC 29042-P74